
1. [bookmark: _Toc122434485][bookmark: _Toc86399947]Overview
This document lists all the changes needed to correct issues in the ATS iWD-TTCN3-B2020-09_D21wk37 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc86399948]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc84540153][bookmark: _Toc86399949]Change 1
	Function name
	f_NR5GC_EmergencyCallWithRegistration()

	Reason for change
	During emergency call establishment proceudure, the PDU SESSION ESTABLISHMENT REQUEST from the UE (step 8) may arrive at the TTCN before the RRCReconfigurationComplete (step 7)  

	Summary of change
	Pass the parameter p_ReceiveForRRCreconfigurationComplete as false in the call to f_NR5GC_RRC_ConnectedState3N_Steps5_8(), so that the PDU SESSION ESTABLISHMENT REQUEST can be handled in f_NR5GC_EmergencyCallPDUSession() if it arrives first (see also change 3).

	TTCN module
	NR5GC_IMS_Procedures.ttcn

	MCC160 Comment
	



Before Change
	  function f_NR5GC_EmergencyCallWithRegistration(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_NR_CellId);
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams);
    
    // Steps 1-3 RRC connection Setup
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (p_NR_CellId, tsc_NR_RRC_TI_Def, emergency);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), v_S_TMSI, '0011'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    // Steps 4-7 Configure SRB2 & DRB
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(p_NR_CellId, v_ReceivedMsg);
    
    // Steps 8-13
    f_NR5GC_EmergencyCallPDUSession(p_NR_CellId);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "IMS Emergency Call Established");
    // Emergency call now active
  }



After Change
	function f_NR5GC_EmergencyCallWithRegistration(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (p_NR_CellId);
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams);
    
    // Steps 1-3 RRC connection Setup
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (p_NR_CellId, tsc_NR_RRC_TI_Def, emergency);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), v_S_TMSI, '0011'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    // Steps 4-7 Configure SRB2 & DRB
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(p_NR_CellId, v_ReceivedMsg, -, -, -, -, -, -, false);
    
    // Steps 8-13
    f_NR5GC_EmergencyCallPDUSession(p_NR_CellId);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "IMS Emergency Call Established");
    // Emergency call now active
  }



1. [bookmark: _Toc86399950]Change 2
	Function name
	f_NR5GC_ModifyPDUSession_AddVoice()

	Reason for change
	During emergency call establishment proceudure, the PDU SESSION MODIFICATION COMPLETE from the UE (step 13) may arrive at the TTCN before the RRCReconfigurationComplete (step 12)  

	Summary of change
	Add interleave block such that steps 12 and 13 can be handled in either order/

	TTCN module
	NR5GC_IMS_Procedures.ttcn

	MCC160 Comment
	



Before Change
	function f_NR5GC_ModifyPDUSession_AddVoice (NR_CellId_Type p_NR_CellId, GSM_MobilityInfo_Type p_GSM_MobilityInfo) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var boolean v_InterworkWithEPS := false;
    var O1_Type v_EPS_BearerId := '00'O;
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_SS_Config;
    var integer v_NewDRBId;
    var boolean v_IsFR1 := f_NR_CellInfo_GetIsFR1(p_NR_CellId);
    var template (value) QosFlow_Identification_Type v_QosFlowId;

    if (isvalue(p_GSM_MobilityInfo.EPS_Bearer)) {  // This is only set if the UE supports EPS
      v_InterworkWithEPS := true;
      v_EPS_BearerId := hex2oct(int2hex(valueof(f_GetNextEPSBearerId (p_GSM_MobilityInfo, v_GMM_MobilityInfo)), 1) & '0'H); // @sic R5-216161 sic@
    }

    // build components of Reconfiguration message
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand(p_GSM_MobilityInfo.SessionId,
                                                                     omit,
                                                                     -, -, -, -,
                                                                     f_BuildDefaultQoSRules("7_Voice", '7A'O), // @sic R5s210073 sic@
                                                                     cs_QoSFlowDescr({cs_QoS_Flow5(v_InterworkWithEPS, v_EPS_BearerId)})); // @sic R5-216161 sic@

    v_NewDRBId := f_NR_GetNextDRBId(p_GSM_MobilityInfo, false);
    // @sic R5-211171 sic@
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(-,
                                                          {cs_NR_DRB_ToAddMod(cs_NR_DRB_SDAP(cs_SDAP_Config(oct2int(p_GSM_MobilityInfo.SessionId), false, {7})), // QFI of 7 in flow 5
                                                                              v_NewDRBId,
                                                                              -, -,
                                                                              cs_38508_PDCP_Config (-, -, -, ms80, omit))}); // UM as per 38.508-1 Table 4.6.3-99: PDCP-Config  @sic R5-213445 sic@
    v_SS_Config := {f_NR_GetRadioBearerConfig_FullUM(p_NR_CellId, v_NewDRBId, -, -, f_ReturnNR_SDAP_Configuration_FromSDAP_Config(valueof(cs_SDAP_Config(oct2int(p_GSM_MobilityInfo.SessionId), false, {7}))))}; // @sic R5-211371, R5s210330 sic@
    v_CellGroupConfig := f_NR_GetCellgroupConfig_DRB(p_NR_CellId, {v_NewDRBId}, false);
    v_CellGroupConfig.rlc_BearerToAddModList := {cs_38508_RLC_BearerConfig_DRB(v_NewDRBId,
                                                                             cs_38508_RLC_Config_UM_DRB(f_NR_UL_UM_RLC(), f_NR_DL_UM_RLC(v_IsFR1)),
                                                                             cs_38508_LogicalChannelConfig_DRB)};

    v_QosFlowId := cs_IP_QosFlowId(oct2int(p_GSM_MobilityInfo.SessionId), 7); // @sic R5s210363 sic@
    f_IP_RtpRtcpLoopback(IP, p_GSM_MobilityInfo.PdnIndex, tsc_IMS_MediaPort_M1, cs_DrbInfo_NR(p_NR_CellId, v_QosFlowId));  // @sic R5s210363 sic@

    // then send RRC Reconfiguration message
    f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                       v_RadioBearerConfig,
                                       v_SS_Config,
                                       v_CellGroupConfig,
                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) });
    SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB2,
                                      cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp;
    if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedAsp.Signalling.Nas[0].Pdu, p_GSM_MobilityInfo.SessionId)) {
    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
    } else {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Speech Call Establishment");
    }
  }
  



After Change
	function f_NR5GC_ModifyPDUSession_AddVoice(NR_CellId_Type p_NR_CellId, GSM_MobilityInfo_Type p_GSM_MobilityInfo) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var boolean v_InterworkWithEPS := false;
    var O1_Type v_EPS_BearerId := '00'O;
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_SS_Config;
    var integer v_NewDRBId;
    var boolean v_IsFR1 := f_NR_CellInfo_GetIsFR1(p_NR_CellId);
    var template (value) QosFlow_Identification_Type v_QosFlowId;

    if (isvalue(p_GSM_MobilityInfo.EPS_Bearer)) {  // This is only set if the UE supports EPS
      v_InterworkWithEPS := true;
      v_EPS_BearerId := hex2oct(int2hex(valueof(f_GetNextEPSBearerId (p_GSM_MobilityInfo, v_GMM_MobilityInfo)), 1) & '0'H); // @sic R5-216161 sic@
    }

    // build components of Reconfiguration message
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand(p_GSM_MobilityInfo.SessionId,
                                                                     omit,
                                                                     -, -, -, -,
                                                                     f_BuildDefaultQoSRules("7_Voice", '7A'O), // @sic R5s210073 sic@
                                                                     cs_QoSFlowDescr({cs_QoS_Flow5(v_InterworkWithEPS, v_EPS_BearerId)})); // @sic R5-216161 sic@

    v_NewDRBId := f_NR_GetNextDRBId(p_GSM_MobilityInfo, false);
    // @sic R5-211171 sic@
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(-,
                                                          {cs_NR_DRB_ToAddMod(cs_NR_DRB_SDAP(cs_SDAP_Config(oct2int(p_GSM_MobilityInfo.SessionId), false, {7})), // QFI of 7 in flow 5
                                                                              v_NewDRBId,
                                                                              -, -,
                                                                              cs_38508_PDCP_Config (-, -, -, ms80, omit))}); // UM as per 38.508-1 Table 4.6.3-99: PDCP-Config  @sic R5-213445 sic@
    v_SS_Config := {f_NR_GetRadioBearerConfig_FullUM(p_NR_CellId, v_NewDRBId, -, -, f_ReturnNR_SDAP_Configuration_FromSDAP_Config(valueof(cs_SDAP_Config(oct2int(p_GSM_MobilityInfo.SessionId), false, {7}))))}; // @sic R5-211371, R5s210330 sic@
    v_CellGroupConfig := f_NR_GetCellgroupConfig_DRB(p_NR_CellId, {v_NewDRBId}, false);
    v_CellGroupConfig.rlc_BearerToAddModList := {cs_38508_RLC_BearerConfig_DRB(v_NewDRBId,
                                                                             cs_38508_RLC_Config_UM_DRB(f_NR_UL_UM_RLC(), f_NR_DL_UM_RLC(v_IsFR1)),
                                                                             cs_38508_LogicalChannelConfig_DRB)};

    v_QosFlowId := cs_IP_QosFlowId(oct2int(p_GSM_MobilityInfo.SessionId), 7); // @sic R5s210363 sic@
    f_IP_RtpRtcpLoopback(IP, p_GSM_MobilityInfo.PdnIndex, tsc_IMS_MediaPort_M1, cs_DrbInfo_NR(p_NR_CellId, v_QosFlowId));  // @sic R5s210363 sic@

    // then send RRC Reconfiguration message
    f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                       v_RadioBearerConfig,
                                       v_SS_Config,
                                       v_CellGroupConfig,
                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) }, -, -, false);

interleave { 
      [] SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete)) {}

    [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB2,
              cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp
  }
    if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedAsp.Signalling.Nas[0].Pdu, p_GSM_MobilityInfo.SessionId)) {
    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
    } else {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Speech Call Establishment");
    }
  }
  




1. [bookmark: _Toc86399951]Change 3
	Function name
	altstep a_NR5GC_AdditionalPDUSessionEstablishmentReq

	Reason for change
	During emergency call establishment proceudure, the PDU SESSION ESTABLISHMENT REQUEST from the UE (step 8) may arrive at the TTCN before the RRCReconfigurationComplete (step 7)  

	Summary of change
	Allow RRCReconfigurationComplete to be received while waiting for PDU SESION ESTABLISHMENT REQUEST

	TTCN module
	NR5GC_NASSteps.ttc

	MCC160 Comment
	



Before Change
	  
altstep a_NR5GC_AdditionalPDUSessionEstablishmentReq(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, ?, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
      {
        f_NR5GC_MsgInDefault_StorePDUSessionEstReq (v_ReceivedAsp);
        repeat;
           
  }



After Change
	altstep a_NR5GC_AdditionalPDUSessionEstablishmentReq(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, ?, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
      {
        f_NR5GC_MsgInDefault_StorePDUSessionEstReq (v_ReceivedAsp);
        repeat;
      }
    [] SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,tsc_NR_RbId_SRB1 , cr_38508_RRCReconfigurationComplete))
    {
      repeat;
    }
      
  }



